********************************************************************************
clear all
set more off
capture log close

********************************************************************************
** File paths
** Set the current directory to the same folder where the code is

global input "../input" 
global output "../output" 
global temp "../temp"

********************************************************************************
* Start log, save it into temp folder
log using "$temp/regressions_cross_section", replace

********************************************************************************
use $temp/data_cross_section, clear

drop if vLanguage == "nl"

********************************************************************************
** Completeness Table
********************************************************************************

gen temp = .

foreach i in 0 1 {

local lVarList "vRelatLength_2014Sep_En vSimilarityEng_2014Sep completeness_2 "
estpost sum `lVarList' if vTreatmentGroup == `i' 
matrix vA`i' = e(mean)
matrix list vA`i'

local lVarList "vRelatLength_2014Sep_Es vSimilarity_2014Sep temp "
estpost sum `lVarList' if vTreatmentGroup == `i'
matrix vB`i' = e(mean)
matrix list vB`i'

local lVarList "vRelatLength_2014Aug_En vSimilarityEng_2014Aug completeness_1 "
estpost sum `lVarList' if vTreatmentGroup == `i'
matrix vC`i' = e(mean)
matrix list vC`i'

local lVarList "vRelatLength_2014Aug_Es vSimilarity_2014Aug temp "
estpost sum `lVarList' if vTreatmentGroup == `i'
matrix vD`i' = e(mean)
matrix list vD`i'

}

matrix mResults = vA1', vA0', vB1', vB0'
matrix list mResults
estadd matrix mResults = mResults, replace

esttab e(mResults, fmt(%9.1f)) using $output/tCompleteness.tex, replace label noobs plain compress fragment unstack varwidth(60) collabels(none) nomtitles coeflabels(vRelatLength_2014Sep_En "Relative length" vSimilarityEng_2014Sep "Similarity" completeness_2 "Completeness" ) prehead(`" & \multicolumn{2}{c}{Compared to English} & \multicolumn{2}{c}{Compared to Spanish}  \\ &  Treatment &  Control &  Treatment &  Control \\ & (1) & (2) & (3) & (4) \\\hline "') posthead(`" & \multicolumn{4}{c}{Panel A: After treatment (2014 September)} \\ "')

matrix mResults = vC1', vC0', vD1', vD0' 
matrix list mResults
estadd matrix mResults = mResults, replace

esttab e(mResults, fmt(%9.1f)) using $output/tCompleteness.tex, append label noobs plain compress fragment unstack varwidth(60) collabels(none) nomtitles coeflabels(vRelatLength_2014Aug_En "Relative length" vSimilarityEng_2014Aug "Similarity" completeness_1 "Completeness" ) posthead(`" \hline & \multicolumn{4}{c}{Panel B: Before treatment (2014 August)} \\ "')

drop temp*

********************************************************************************
** Quality vs length
********************************************************************************
** Overall quality
xtile tempPercentilesOverall = overall_1, nquantiles(5)
tabstat overall_1, by(tempPercentilesOverall)
graph box vLogLength2014Aug, over(tempPercentilesOverall) nooutsides graphregion(fcolor(gs16)) ytitle("Log. length") b1title("Quality") 
graph export $output/gLength_overall.pdf, replace 
drop temp*

** Completeness
xtile tempPercentilesComplete = completeness_1, nquantiles(5)
tabstat completeness_1, by(tempPercentilesComplete)
graph box vLogLength2014Aug, over(tempPercentilesComplete) nooutsides graphregion(fcolor(gs16)) ytitle("Log. length") b1title("Completeness (compared to English)") 
graph export $output/gLength_completeness.pdf, replace 
drop temp*

** Similarity
xtile tempPercentilesComplete = completeness_1, nquantiles(5)
tabstat completeness_1, by(tempPercentilesComplete)
graph box vSimilarity_2014Sep, over(tempPercentilesComplete) nooutsides graphregion(fcolor(gs16)) ytitle("Similarity to Spanish") b1title("Completeness (compared to English)") 
graph export $output/gSimilarity_completeness.pdf, replace 
drop temp*

** Similarity
xtile tempPercentiles = vSimilarity_2014Sep, nquantiles(5)
tabstat vSimilarity_2014Sep, by(tempPercentiles)
graph box vLogLength2014Aug, over(tempPercentiles) nooutsides graphregion(fcolor(gs16)) ytitle("Log. length") b1title("Similarity to Spanish") 
graph export $output/gLength_similarity.pdf, replace 
drop temp*

********************************************************************************
** Balance tests: Predicting treatment status with pre-treatment variables
********************************************************************************

local lVarList " vLogLength2014Aug overall_1 completeness_1 interesting_1 wellwritten_1 illustrated_1 vSimilarity_2014Aug vAverUsersPreTreatment vAverEditDaysPreTreatment vAverEditDistPreTreatment vAverEditDistDCapPreTreatment  "

estpost sum `lVarList'if vTreatmentGroup==0
matrix mB=e(mean)
estpost sum `lVarList' if vTreatmentGroup == 1
matrix mC=e(mean)
estpost ttest `lVarList' , by(vTreatmentGroup)
matrix mD=e(p)

foreach var of varlist `lVarList' {
	ranksum `var' , by(vTreatmentGroup)
	scalar sZ = r(z)
	scalar list sZ
	scalar sP=2*(1-normal(abs(sZ)))
	scalar list sP
	matrix mE = nullmat(mE) , sP
	matrix list mE
}

estpost sum `lVarList' 
matrix mF=e(count)

matrix list mB
matrix list mE
matrix list mF

matrix mKapsas = mB',mC',mD',mE',mF'
estadd matrix mResults = mKapsas, replace

** Nice formatting of the table 
esttab e(mResults, fmt(3 3 %9.3f %9.3f 0)) using $output/tBalanceTest.tex, label nonum noobs plain compress fragment unstack varwidth(60)  collabels(none) nomtitles prehead(`" & Control & Treatment & t-test & Wilcoxon test & Obs. \\ & group &  group & p-value & p-value & \\ & mean &  mean &  &  & \\  & (1) & (2) & (3) & (4) & (5) \\ \hline "') replace

********************************************************************************
** Main regression table
********************************************************************************
foreach var of varlist vDifLogLength_20182014 vDif_overall_32 vDifSimilarity_20182014 {
	regress `var' vTreatmentGroup i.vNumLanguage i.vNumPage
	estimates store `var'
	sum `var' 
	estadd scalar p_mean = `r(mean)'
	estadd scalar p_sd = `r(sd)'
}

estout vDifLogLength_20182014 vDif_overall_32 vDifSimilarity_20182014 using $output/tRegrLengthQuality.tex, keep(vTreatmentGroup) cells(b(star fmt(3)) se(par fmt(3))) style(tex) mlabels(none) collabels(none) label wrap starlevels(* .1 ** .05 *** .01) stats(p_mean p_sd r2_a N , fmt(3 3 3 %9.0g ) labels("Mean dep. var." "SD dep. var." "Adj. R-squared" "Observations")) indicate("Language FE = *vNumLanguage*" "City FE = *vNumPage*" ) prehead(`" & \multicolumn{3}{c}{Change in page length or quality (\$y_{2018Sep} - y_{2014Sep}\$)} \\ & \$\Delta\$Log. page length & \$\Delta\$Quality rating & \$\Delta\$Similarity to Spanish\\  & (1) & (2) & (3)  \\ \hline "') replace

********************************************************************************
** Regressions: detailed quality measures
********************************************************************************
foreach p in completeness interesting wellwritten illustrated {
	regress vDif_`p'_32 vTreatmentGroup i.vNumLanguage i.vNumPage
	estimates store e`p'
	sum vDif_`p'_32
	estadd scalar p_mean = `r(mean)'
	estadd scalar p_sd = `r(sd)'
}

estout ecompleteness einteresting eillustrated ewellwritten using $output/tRegrManualRatingsDetailed.tex, keep(vTreatmentGroup) cells(b(star fmt(3)) se(par fmt(3))) style(tex) mlabels(none) collabels(none) label wrap starlevels(* .1 ** .05 *** .01) stats(p_mean p_sd r2_a N , fmt(3 3 3 %9.0g ) labels("Mean dep. var." "SD dep. var." "Adj. R-squared" "Observations")) indicate("Language FE = *vNumLanguage*" "City FE = *vNumPage*" ) prehead(`"  & \multicolumn{4}{c}{Change in page quality (\$y_{2018Sep} - y_{2014Sep}\$)} \\ & \$\Delta\$Complete & \$\Delta\$Interesting & \$\Delta\$Illustrated & \$\Delta\$Well-written \\  & (1) & (2) & (3) & (4) \\ \hline "') replace

********************************************************************************
** Heterogeneity: Main regression table
********************************************************************************

program pRegressionsHeterogeneity
{
local lRegrHeterog $gRegrHeterog
sum `lRegrHeterog', detail
display "`r(p50)'"
gen vLow = (`lRegrHeterog'<`r(p50)')	
foreach var of varlist vDifLogLength_20182014 vDif_overall_32 vDifSimilarity_20182014 {
	sum `lRegrHeterog', detail
	display "`r(p50)'"
	gen vInterTreatLow = vTreatmentGroup*(`lRegrHeterog'<`r(p50)')
	regress `var' vTreatmentGroup vInterTreatLow vLow i.vNumLanguage i.vNumPage
	estimates store `var'
	sum `var' 
	estadd scalar p_mean = `r(mean)'
	estadd scalar p_sd = `r(sd)'
	drop vInterTreatLow
}
drop vLow
}
end

** By overall
global gRegrHeterog "overall_1"
pRegressionsHeterogeneity

estout vDifLogLength_20182014 vDif_overall_32 vDifSimilarity_20182014 using $output/tRobustLengthQuality_HeterQuality.tex, keep(vTreatmentGroup vInterTreatLow vLow) cells(b(star fmt(3)) se(par fmt(3))) style(tex) mlabels(none) collabels(none) label wrap starlevels(* .1 ** .05 *** .01) stats(p_mean p_sd r2_a N , fmt(3 3 3 %9.0g ) labels("Mean dep. var." "SD dep. var." "Adj. R-squared" "Observations")) indicate("Language FE = *vNumLanguage*" "City FE = *vNumPage*" ) varlabel(vLow "Below median" vInterTreatLow "Treatment group $\times$ Below median") prehead(`" & \multicolumn{3}{c}{Change in page length or quality (\$y_{2018Sep} - y_{2014Sep}\$)} \\ & \$\Delta\$Log. page length & \$\Delta\$Quality rating & \$\Delta\$Similarity to Spanish\\  & (1) & (2) & (3)  \\ \hline & \multicolumn{3}{c}{Panel A: Heterogeneity by page quality} \\ "') replace

** By completeness
global gRegrHeterog "completeness_1"
pRegressionsHeterogeneity

estout vDifLogLength_20182014 vDif_overall_32 vDifSimilarity_20182014 using $output/tRobustLengthQuality_HeterQuality.tex, keep(vTreatmentGroup vInterTreatLow vLow) cells(b(star fmt(3)) se(par fmt(3))) style(tex) mlabels(none) collabels(none) label wrap starlevels(* .1 ** .05 *** .01) stats(p_mean p_sd r2_a N , fmt(3 3 3 %9.0g ) labels("Mean dep. var." "SD dep. var." "Adj. R-squared" "Observations")) indicate("Language FE = *vNumLanguage*" "City FE = *vNumPage*" ) varlabel(vLow "Below median" vInterTreatLow "Treatment group $\times$ Below median") prehead(`" \hline & \multicolumn{3}{c}{Panel B: Heterogeneity by page completeness} \\ "') append

** By length
global gRegrHeterog "vRelatLength_2014Aug_Es"
pRegressionsHeterogeneity

estout vDifLogLength_20182014 vDif_overall_32 vDifSimilarity_20182014 using $output/tRobustLengthQuality_HeterLengthAge.tex, keep(vTreatmentGroup vInterTreatLow vLow) cells(b(star fmt(3)) se(par fmt(3))) style(tex) mlabels(none) collabels(none) label wrap starlevels(* .1 ** .05 *** .01) stats(p_mean p_sd r2_a N , fmt(3 3 3 %9.0g ) labels("Mean dep. var." "SD dep. var." "Adj. R-squared" "Observations")) indicate("Language FE = *vNumLanguage*" "City FE = *vNumPage*" ) varlabel(vLow "Below median" vInterTreatLow "Treatment group $\times$ Below median") prehead(`" & \multicolumn{3}{c}{Change in page length or quality (\$y_{2018Sep} - y_{2014Sep}\$)} \\ & \$\Delta\$Log. page length & \$\Delta\$Quality rating & \$\Delta\$Similarity to Spanish\\  & (1) & (2) & (3)  \\ \hline & \multicolumn{3}{c}{Panel A: Heterogeneity by page relative length (to Spanish)} \\ "') replace

** By length
global gRegrHeterog "vPageAge2014Aug"
pRegressionsHeterogeneity

estout vDifLogLength_20182014 vDif_overall_32 vDifSimilarity_20182014 using $output/tRobustLengthQuality_HeterLengthAge.tex, keep(vTreatmentGroup vInterTreatLow vLow) cells(b(star fmt(3)) se(par fmt(3))) style(tex) mlabels(none) collabels(none) label wrap starlevels(* .1 ** .05 *** .01) stats(p_mean p_sd r2_a N , fmt(3 3 3 %9.0g ) labels("Mean dep. var." "SD dep. var." "Adj. R-squared" "Observations")) indicate("Language FE = *vNumLanguage*" "City FE = *vNumPage*" ) varlabel(vLow "Below median" vInterTreatLow "Treatment group $\times$ Below median") prehead(`" \hline & \multicolumn{3}{c}{Panel B: Heterogeneity by page age} \\ "') append

********************************************************************************
** Robustness: main table
********************************************************************************
foreach var of varlist vDifLogLength_20182014 vDif_overall_32 vDifSimilarity_20182014 {
	regress `var' vTreatmentGroup 
	estimates store `var'_1
	sum `var' 
	estadd scalar p_mean = `r(mean)'
	estadd scalar p_sd = `r(sd)'

	regress `var' vTreatmentGroup i.vGroup 
	estimates store `var'_2
	sum `var' 
	estadd scalar p_mean = `r(mean)'
	estadd scalar p_sd = `r(sd)'

	regress `var' vTreatmentGroup i.vNumLanguage 
	estimates store `var'_3
	sum `var' 
	estadd scalar p_mean = `r(mean)'
	estadd scalar p_sd = `r(sd)'
}

estout vDifLogLength_20182014_1 vDifLogLength_20182014_2 vDifLogLength_20182014_3 vDif_overall_32_1 vDif_overall_32_2 vDif_overall_32_3 vDifSimilarity_20182014_1 vDifSimilarity_20182014_2 vDifSimilarity_20182014_3 using $output/tRobustRegrLengthQuality_Controls.tex, keep(vTreatmentGroup) cells(b(star fmt(3)) se(par fmt(3))) style(tex) mlabels(none) collabels(none) label wrap starlevels(* .1 ** .05 *** .01) stats(p_mean p_sd N , fmt(3 3  %9.0g ) labels("Mean dep. var." "SD dep. var." "Observations")) indicate("Group FE = *vGroup*" "Language FE = *vNumLanguage*" ) prehead(`" & \multicolumn{9}{c}{Change in page length or quality (\$y_{2018Sep} - y_{2014Sep}\$)} \\ & \multicolumn{3}{c}{\$\Delta\$Log. page length} & \multicolumn{3}{c}{\$\Delta\$Quality rating} & \multicolumn{3}{c}{\$\Delta\$Similarity to Spanish} \\  & (1) & (2) & (3) & (4) & (5) & (6) & (7) & (8) & (9) \\ \hline "') replace

********************************************************************************
** Distributions before treatment
********************************************************************************
foreach var of varlist vLogLength2014Aug overall_1 completeness_1 interesting_1 wellwritten_1 illustrated_1 vSimilarity_2014Aug vAverUsersPreTreatment vAverEditDaysPreTreatment vAverEditDistPreTreatment vAverEditDistDCapPreTreatment {

	kdensity `var' if vTreatmentGroup==1 , lcolor(red) addplot(kdensity `var' if vTreatmentGroup==0, lcolor(black)) kernel(epan) graphregion(style(none)) scheme(s1mono) legend(label(1 "Treatment") label(2 "Control"))
	graph export $output/gBC_`var'.pdf, replace 

}

********************************************************************************
** Graphs of quality change
********************************************************************************
preserve

local lQuality overall

keep vDif_`lQuality'_21 vDif_`lQuality'_32 vTreatmentGroup vPage vLanguage
reshape long vDif_`lQuality'_, i(vTreatmentGroup vPage vLanguage) j(vTime)

statsby mean=r(mean) ub=r(ub) lb=r(lb) N=r(N), by(vTreatmentGroup vTime) clear: ci mean vDif_`lQuality'_
gen vInterTreatTime = vTreatmentGroup if vTime == 21
replace vInterTreatTime = vTreatmentGroup+3 if vTime == 32

twoway (bar mean vInterTreatTime if vTreatmentGroup==0, color(gold) lcolor(gs0)) (bar mean vInterTreatTime if vTreatmentGroup==1, color(dkgreen) lcolor(gs0)) (rcap ub lb vInterTreatTime, color(gs0)), legend(row(1) order(1 "Control" 2 "Treatment") size(medlarge)) xlabel(0.5 "Aug 2014" 3.5 "Sep 2014 - 2018" , noticks labsize(medlarge)) ytitle("{&Delta} Overall quality", size(medlarge)) xtitle("") graphregion(fcolor(gs16)) scheme(s2mono)
graph export $output/gChange_Quality_`lQuality'.pdf, replace 

restore

preserve

local lQuality completeness

keep vDif_`lQuality'_21 vDif_`lQuality'_32 vTreatmentGroup vPage vLanguage
reshape long vDif_`lQuality'_, i(vTreatmentGroup vPage vLanguage) j(vTime)

statsby mean=r(mean) ub=r(ub) lb=r(lb) N=r(N), by(vTreatmentGroup vTime) clear: ci mean vDif_`lQuality'_
gen vInterTreatTime = vTreatmentGroup if vTime == 21
replace vInterTreatTime = vTreatmentGroup+3 if vTime == 32

twoway (bar mean vInterTreatTime if vTreatmentGroup==0, color(gold) lcolor(gs0)) (bar mean vInterTreatTime if vTreatmentGroup==1, color(dkgreen) lcolor(gs0)) (rcap ub lb vInterTreatTime, color(gs0)), legend(row(1) order(1 "Control" 2 "Treatment") size(medlarge)) xlabel(0.5 "Aug 2014" 3.5 "Sep 2014 - 2018" , noticks labsize(medlarge)) ytitle("{&Delta} Completeness", size(medlarge)) xtitle("") graphregion(fcolor(gs16)) scheme(s2mono)
graph export $output/gChange_Quality_`lQuality'.pdf, replace 

restore

preserve

local lQuality interesting

keep vDif_`lQuality'_21 vDif_`lQuality'_32 vTreatmentGroup vPage vLanguage
reshape long vDif_`lQuality'_, i(vTreatmentGroup vPage vLanguage) j(vTime)

statsby mean=r(mean) ub=r(ub) lb=r(lb) N=r(N), by(vTreatmentGroup vTime) clear: ci mean vDif_`lQuality'_
gen vInterTreatTime = vTreatmentGroup if vTime == 21
replace vInterTreatTime = vTreatmentGroup+3 if vTime == 32

twoway (bar mean vInterTreatTime if vTreatmentGroup==0, color(gold) lcolor(gs0)) (bar mean vInterTreatTime if vTreatmentGroup==1, color(dkgreen) lcolor(gs0)) (rcap ub lb vInterTreatTime, color(gs0)), legend(row(1) order(1 "Control" 2 "Treatment") size(medlarge)) xlabel(0.5 "Aug 2014" 3.5 "Sep 2014 - 2018" , noticks labsize(medlarge)) ytitle("{&Delta} Interesting", size(medlarge)) xtitle("") graphregion(fcolor(gs16)) scheme(s2mono)
graph export $output/gChange_Quality_`lQuality'.pdf, replace 

restore

preserve

local lQuality wellwritten

keep vDif_`lQuality'_21 vDif_`lQuality'_32 vTreatmentGroup vPage vLanguage
reshape long vDif_`lQuality'_, i(vTreatmentGroup vPage vLanguage) j(vTime)

statsby mean=r(mean) ub=r(ub) lb=r(lb) N=r(N), by(vTreatmentGroup vTime) clear: ci mean vDif_`lQuality'_
gen vInterTreatTime = vTreatmentGroup if vTime == 21
replace vInterTreatTime = vTreatmentGroup+3 if vTime == 32

twoway (bar mean vInterTreatTime if vTreatmentGroup==0, color(gold) lcolor(gs0)) (bar mean vInterTreatTime if vTreatmentGroup==1, color(dkgreen) lcolor(gs0)) (rcap ub lb vInterTreatTime, color(gs0)), legend(row(1) order(1 "Control" 2 "Treatment") size(medlarge)) xlabel(0.5 "Aug 2014" 3.5 "Sep 2014 - 2018" , noticks labsize(medlarge)) ytitle("{&Delta} Well-written", size(medlarge)) xtitle("") graphregion(fcolor(gs16)) scheme(s2mono)
graph export $output/gChange_Quality_`lQuality'.pdf, replace 

restore

preserve

local lQuality illustrated

keep vDif_`lQuality'_21 vDif_`lQuality'_32 vTreatmentGroup vPage vLanguage
reshape long vDif_`lQuality'_, i(vTreatmentGroup vPage vLanguage) j(vTime)

statsby mean=r(mean) ub=r(ub) lb=r(lb) N=r(N), by(vTreatmentGroup vTime) clear: ci mean vDif_`lQuality'_
gen vInterTreatTime = vTreatmentGroup if vTime == 21
replace vInterTreatTime = vTreatmentGroup+3 if vTime == 32

twoway (bar mean vInterTreatTime if vTreatmentGroup==0, color(gold) lcolor(gs0)) (bar mean vInterTreatTime if vTreatmentGroup==1, color(dkgreen) lcolor(gs0)) (rcap ub lb vInterTreatTime, color(gs0)), legend(row(1) order(1 "Control" 2 "Treatment") size(medlarge)) xlabel(0.5 "Aug 2014" 3.5 "Sep 2014 - 2018" , noticks labsize(medlarge)) ytitle("{&Delta} Illustrated", size(medlarge)) xtitle("") graphregion(fcolor(gs16)) scheme(s2mono)
graph export $output/gChange_Quality_`lQuality'.pdf, replace 

restore

** Figure for similarity
preserve

keep vDifSimilarity_2014SepAug vDifSimilarity_20182014 vTreatmentGroup vPage vLanguage
rename vDifSimilarity_2014SepAug vDifSimilarity_21
rename vDifSimilarity_20182014 vDifSimilarity_32
reshape long vDifSimilarity_, i(vTreatmentGroup vPage vLanguage) j(vTime) 

statsby mean=r(mean) ub=r(ub) lb=r(lb) N=r(N), by(vTreatmentGroup vTime) clear: ci mean vDifSimilarity_
gen vInterTreatTime = vTreatmentGroup if vTime == 21
replace vInterTreatTime = vTreatmentGroup+3 if vTime == 32

twoway (bar mean vInterTreatTime if vTreatmentGroup==0, color(gold) lcolor(gs0)) (bar mean vInterTreatTime if vTreatmentGroup==1, color(dkgreen) lcolor(gs0)) (rcap ub lb vInterTreatTime, color(gs0)), legend(row(1) order(1 "Control" 2 "Treatment") size(medlarge)) xlabel(0.5 "Aug 2014" 3.5 "Sep 2014 - 2018" , noticks labsize(medlarge)) ytitle("{&Delta} Similarity to Spanish", size(medlarge)) xtitle("") graphregion(fcolor(gs16)) scheme(s2mono)
graph export $output/gChange_Similarity.pdf, replace 

restore

********************************************************************************
** With Dutch 
********************************************************************************
use $temp/data_cross_section, clear

tabulate vLanguage vTreatmentGroup

regress vDifLogLength_20182014 vTreatmentGroup i.vNumLanguage i.vNumPage
estimates store eITT
sum vDifLogLength_20182014
estadd scalar p_mean = `r(mean)'
estadd scalar p_sd = `r(sd)'

replace vTreatmentGroup = 0 if vLanguage == "nl"

regress vDifLogLength_20182014 vTreatmentGroup i.vNumLanguage i.vNumPage
estimates store eDutch
sum vDifLogLength_20182014
estadd scalar p_mean = `r(mean)'
estadd scalar p_sd = `r(sd)'

estout  eDutch eITT using $output/tRobustRegrLengthQuality_Dutch.tex, keep(vTreatmentGroup) cells(b(star fmt(3)) se(par fmt(3))) style(tex) mlabels(none) collabels(none) label wrap starlevels(* .1 ** .05 *** .01) stats(p_mean p_sd r2_a N , fmt(3 3 3 %9.0g ) labels("Mean dep. var." "SD dep. var." "Adj. R-squared" "Observations")) indicate("Language FE = *vNumLanguage*" "City FE = *vNumPage*" ) prehead(`" & \multicolumn{2}{c}{Change in page log. length (\$y_{2018Sep} - y_{2014Sep}\$)} \\ & Dutch in control gr. & Intention to Treat  \\ & (1) & (2)  \\ \hline "') replace

********************************************************************************
log close
clear
